AE341 Modeling and Analysis Lab
Exercise 3

Kinematic Analysis of the A-frame mechanism

Ayush Sharma (SC19B008), G. Ramana Bharathi (SC19B028)
Department of Aerospace Engineering
Indian Institute of Space Science and Technology

(Dated: October 18, 2021)

In the previous exercise, we had proposed a mathematical model for the Dragon capsule, but we did not analyse the A-frame structure in detail. In this exercise, our focus is to do the kinematic analysis for the A-frame structure, the purpose of which is to study the motion (position, velocity and acceleration) of the structure, without considering the forces that cause the corresponding motion. We first did a kinematic analysis for a planar 4-bar mechanism, since it closely resembles the A-frame structure, and then made a few modifications to obtain the kinematic analysis results for the A-frame structure in question.

I. INTRODUCTION

Kinematic analysis of a structure constitutes of the position analysis, velocity analysis and acceleration analysis of the structure. We write loop closure equations(for position analysis) in terms of the fixed parameters, the input parameters and the unknowns, and then solve for the unknowns. The solution to the equations yield the angular position of each of the links of the mechanism, using which it is relatively simple to find the exact location of any point on any link, given the fixed and input parameters. For obtaining the velocity and acceleration equations, we differentiate the components of the loop closure equation once and twice respectively and then solve for the unknowns to perform the required analysis.

II. THEORY

A schematic diagram for a planar 4-bar mechanism has been provided in figure 1. The degrees of freedom for the mechanism can be computed using the following formula:

$$\boxed{DoF = 3(n-1) - 2j_1 - j_2}$$

where n is the number of links, j_1 is the number of joints with 1 relative degree of freedom, and j_2 is the number of joints with 2 relative degrees of freedom. In the case of the 4-bar mechanism, there are 4 links, and 4 revolute joints (all of which are joints with 1 relative DoF). Hence, we have

- n=4
- j_1=4
- j_2=0

which means that the planar 4-bar mechanism is a **1 DoF** mechanism. Generally, for a 4-bar mechanism, link AB is called the input link, link BC is called the coupler, and link CD is called the output link. The explanations for the position, velocity and acceleration analyses of the mechanism are provided in the following sections:



FIG. 1. Schematic of the A-frame structure as seen from the side

II.1. Position Analysis

1. **Co-ordinate Frame :** The reference frame x-axis passes through the line connecting the joints A and D, while the y-axis is perpendicular to the x-axis and lies in the plane of the paper.
2. **Fixed Parameters :** Lengths AB, BC, CD, DA, angle θ_1(which is the angle made by the link DA with the positive x-axis)
3. **Input Parameter :** In case of **1 DoF** systems, we require only 1 input parameter. In case of position

analysis, we have θ_2 as the input parameter.

4. **Loop Closure Equations :** We are considering the loop **ABCDA** for writing the equations. The vector loop closure equation along with the x- and y- component equations are:

$$\overrightarrow{\boldsymbol{AB}} + \overrightarrow{\boldsymbol{BC}} + \overrightarrow{\boldsymbol{CD}} + \overrightarrow{\boldsymbol{DA}} = \overrightarrow{\boldsymbol{O}}$$

$$\begin{gathered} AB\cos\angle AB + BC\cos\angle BC + CD\cos\angle CD \\ + DA\cos\angle DA = 0 \\ AB\sin\angle AB + BC\sin\angle BC + CD\sin\angle CD \\ + DA\sin\angle DA = 0 \end{gathered}$$

From the figure, we can make the following observations:

- $\angle AB = \theta_2$
- $\angle BC = \theta_3$
- $\angle CD = \theta_4 + \pi$
- $\angle DA = \theta_1 = \pi$

The equations can be simplified by keeping the term containing θ_3 on the LHS and all the other terms on the RHS, and then squaring and adding. This would let us get rid of θ_3, and we could compute θ_4 using the following relation:

$$\theta_4 = \arccos\frac{-z}{\sqrt{x^2+y^2}} + \arctan\frac{y}{x}$$

where

$$\begin{aligned} x &= -AB\cos\theta_2 + DA \\ y &= -AB\sin\theta_2 \\ z &= \frac{x^2+y^2+CD^2-BC^2}{2CD} \end{aligned}$$

Then, we can use the components of the loop closure equations to determine the angle θ_3.

II.2. Velocity Analysis

1. **Co-ordinate Frame :** The reference frame x-axis passes through the line connecting the joints A and D, while the y-axis is perpendicular to the x-axis and lies in the plane of the paper.

2. **Fixed Parameters :** Lengths AB, BC, CD, DA, and the positions and orientations of all the links.

3. **Input Parameter :** In case of **1 DoF** systems, we require only 1 input parameter. In case of velocity analysis, we have $\omega_2 = \dot{\theta}_2$ as the input parameter.

4. **Velocity Equations :** The equations for velocity analysis can be obtained easily by just differentiating the loop closure equations with respect to time. The equations hence obtained are (in the matrix form):

$$\begin{bmatrix} BC\sin\theta_3 & -CD\sin\theta_4 \\ BC\cos\theta_3 & -CD\cos\theta_4 \end{bmatrix} \begin{bmatrix} \dot{\theta}_3 \\ \dot{\theta}_4 \end{bmatrix} = \begin{bmatrix} -AB\dot{\theta}_2\sin\theta_2 \\ -AB\dot{\theta}_2\cos\theta_2 \end{bmatrix}$$

Now, the above system of linear equations can be solved analytically with ease using any of the available commercial packages.

II.3. Acceleration Analysis

1. **Co-ordinate Frame :** The reference frame x-axis passes through the line connecting the joints A and D, while the y-axis is perpendicular to the x-axis and lies in the plane of the paper.

2. **Fixed Parameters :** Lengths AB, BC, CD, DA, and the positions and orientations of all the links, the velocities of all the links.

3. **Input Parameter :** In case of **1 DoF** systems, we require only 1 input parameter. In case of acceleration analysis, we have $\alpha_2 = \ddot{\theta}_2$ as the input parameter.

4. **Acceleration Equations :** The equations for acceleration analysis can be obtained easily by doubly differentiating the loop closure equations with respect to time. The equations hence obtained are (in the matrix form):

$$\begin{bmatrix} BC\cos\theta_3 & -CD\cos\theta_4 \\ BC\sin\theta_3 & -CD\sin\theta_4 \end{bmatrix} \begin{bmatrix} \ddot{\theta}_3 \\ \ddot{\theta}_4 \end{bmatrix} =$$

$$\begin{bmatrix} -AB\ddot{\theta}_2\cos\theta_2 + AB\dot{\theta}_2^{\,2}\sin\theta_2 + BC\dot{\theta}_3^{\,2}\sin\theta_3 - CD\dot{\theta}_4^{\,2}\sin\theta_4 \\ -AB\ddot{\theta}_2\sin\theta_2 - AB\dot{\theta}_2^{\,2}\cos\theta_2 - BC\dot{\theta}_3^{\,2}\cos\theta_3 + CD\dot{\theta}_4^{\,2}\cos\theta_4 \end{bmatrix}$$

Now, the above system of linear equations can be solved analytically with ease using any of the available commercial packages.

III. DATA PRESENTATION

This section contains the plots showing the variations of θ_2 (input) [2], θ_3 [3], θ_4 [4], ω_2 [5], ω_3 [6], ω_4 [7], α_2 [8], α_3 [9] and α_4 [10] as a function of time.



FIG. 2. Plot for θ_2 vs time



FIG. 3. Plot for θ_3 vs time



FIG. 4. Plot for θ_4 vs time



FIG. 5. Plot for ω_2 vs time



FIG. 6. Plot for ω_3 vs time



FIG. 7. Plot for ω_4 vs time



FIG. 8. Plot for α_2 vs time



FIG. 9. Plot for α_3 vs time



FIG. 10. Plot for α_4 vs time



FIG. 11. Trajectory of point of suspension of pendulum

IV. RESULTS AND DISCUSSIONS

Parameters used

We use the following fixed parameters and input parameters (θ_2) in the model based on our observation of the A-frame:

$$AB = 5\ m$$

$$BC = 6\ m$$

$$CD = 5\ m$$

$$DA = 6\ m$$

$$DP = 10\ m$$

$$\theta_1 = \pi\ rad$$

$$\theta_2 = \frac{\pi}{6} + \frac{\pi}{2 \times 45^2} t^2$$

Time range $\rightarrow$ 0 to 45 seconds

The analyses have been performed in Julia [1] and python [2]. We obtain the variations of the variables θ_2 (input), θ_3, θ_4, ω_2, ω_3, ω_4, α_2, α_3 and α_4 as a function of time. Since θ_2 is assumed to be a parabolic function of time, we observe the corresponding parabolic relations and its derivatives in the plots for θ_2,ω_2 and α_2. Since the lengths of AB and CD are assumed to be equal, the four-bar is a parallelogram at all times t, and hence, θ_3, ω_3 and α_3 are always all zero as time t varies. The variations in θ_4, ω_4 and α_4 and θ_2, ω_2 and α_2 against time are correspondingly similar to each other because of the parallelogram formation.

The position, velocity and acceleration of any point on the A-frame can now be determined using the analyses performed. The angular velocity and angular acceleration on a rigid bar is constant for every point on the bar and the angular displacement from a fixed origin is offset by the constant value which can be obtained from the geometry of the bar. We can also obtain the trajectory of the point of suspension of the pendulum using:

$$x_p = DA + DP\cos\theta_4$$
$$y_p = DP\sin\theta_4$$

A plot of the trajectory is shown in Fig. 11.

V. CONCLUSION

Kinematic analysis of four-bar mechanism has been performed and the results have been discussed. The planar motion of the A-frame can be represented by the kinematics of the four-bar.

Appendix A: Scripts used for modelling

1. Julia code for performing kinematic analysis of four-bar mechanism

```
"""
Julia script for kinematics analysis of four-bar mechanism.
Contains sections for position, velocity, and acceleration analyses.

"""

using Plots
plotlyjs()


# fixed parameters
AB=5
BC=6
CD=5
DA=6
DP=10
θ₁=π

#time limit
t=0:0.01:45

# position analysis

#independent coordinate
ζ₀=π/6 # angle subtended by the A-frame [rad]
β₀=dζ=90*π/(180*45*45)
θ₂=ζ=ζ₀.+β₀.*t.^2

x=-AB.*cos.(θ₂).-DA.*cos.(θ₁)
y=-AB.*sin.(θ₂).-DA.*sin.(θ₁)

z=(x.^2 .+ y.^2 .+ CD^2 .- BC^2)./(2 .* CD)

θ₄=acos.(-z./sqrt.(x.^2 .+ y.^2)) .+ atan.(y,x)

y_C=CD.*sin.(θ₄)
y_B=AB.*sin.(θ₂)

x_C=CD.*cos.(θ₄).+DA
x_B=AB.*cos.(θ₂)

θ₃=atan.(y_C .- y_B,x_C .- x_B)

#plotting
display(plot(t,θ₂,linewidth=2,xaxis="time (s)",yaxis="θ (rad)",
    label="θ₂",color="blue"))
savefig("./plots/ex3-11.png")
display(plot(t,θ₃,linewidth=2,xaxis="time (s)",yaxis="θ (rad)",
    label="θ₃",color="purple",ylim=[-0.4,0.4]))
savefig("./plots/ex3-12.png")
display(plot(t,θ₄,linewidth=2,xaxis="time (s)",yaxis="θ (rad)",
    label="θ₄",color="orange"))
savefig("./plots/ex3-13.png")

#trajectory of point of suspension
xₚ=DA .+ DP.*cos.(θ₄)
yₚ=DP.*sin.(θ₄)
display(plot(xₚ,yₚ,label="trajectory",xaxis="x (m)",yaxis="y (m)",
    linewidth=2,xlim=[0,15],xticks=0:1:15,ylim=[0,15],yticks=0:1:15))
```

```julia
savefig("./plots/ex3-trajectory.png")


#velocity analysis

#independent coordinate
θ̇₂=dθ₂=2 .*β₀.*t

X=ones(2,length(t))

#solving the equations
for i in 1:length(t)
    A=[BC.*sin.(θ₃[i])  -CD.*sin.(θ₄[i]); BC.*cos.(θ₃[i])  -CD.*cos.(θ₄[i])]
    # X=[θ̇₃;θ̇₄]
    B=[-AB.*θ̇₂[i].*sin.(θ₂[i]); -AB.*θ̇₂[i].*cos.(θ₂[i])]

    Aᴵ=inv(A)
    @assert size(Aᴵ,2)==size(B,1) "dimension mismatch"
    X[:,i]=Aᴵ*B
end

(θ̇₃,θ̇₄)=[X[i,:] for i in 1:size(X,1)]

#plotting
display(plot(t,θ̇₂,linewidth=2,xaxis="time (s)",yaxis="ω (rad/s)",
    label="ω₂",color="blue"))
savefig("./plots/ex3-21.png")
display(plot(t,θ̇₃,linewidth=2,xaxis="time (s)",yaxis="ω (rad/s)",
    label="ω₃",color="purple",ylim=[-0.4,0.4]))
savefig("./plots/ex3-22.png")
display(plot(t,θ̇₄,linewidth=2,xaxis="time (s)",yaxis="ω (rad/s)",
    label="ω₄",color="orange"))
savefig("./plots/ex3-23.png")


#acceleration analysis

#independent coordinate
θ̈₂=dθ̇₂=2*β₀.*ones(length(t))

Y=ones(2,length(t))

#solving the equations
for i in 1:length(t)
    C=[BC*cos(θ₃[i])  -CD*cos(θ₄[i]); BC*sin.(θ₃[i])  -CD*sin.(θ₄[i])]
    # Y=[θ̈₃;θ̈₄]
    D=[(-AB*θ̈₂[i]*cos.(θ₂[i])+AB*(θ̇₂[i]^2)*sin(θ₂[i])+BC*(θ̇₃[i]^2)*sin.(θ₃[i])
            -CD*(θ̇₄[i]^2)*sin(θ₄[i]));
        (-AB*θ̈₂[i]*sin(θ₂[i])-AB*(θ̇₂[i]^2)*cos(θ₂[i])-BC*(θ̇₃[i]^2)*cos.(θ₃[i])
            +CD*(θ̇₄[i]^2)*cos(θ₄[i]))]

            Cᴵ=inv(C)
    @assert size(Cᴵ,2)==size(D,1) "dimension mismatch"
    Y[:,i]=Cᴵ*D
end

(θ̈₃,θ̈₄)=[Y[i,:] for i in 1:size(Y,1)]

#plotting
display(plot(t,θ̈₂,linewidth=2,xaxis="time (s)",yaxis="α (rad/s^2)",
    label="α₂",color="blue",ylim=[0,0.0025]))
savefig("./plots/ex3-31.png")
display(plot(t,θ̈₃,linewidth=2,xaxis="time (s)",yaxis="α (rad/s^2)",
    label="α₃",color="purple",ylim=[-0.4,0.4]))
savefig("./plots/ex3-32.png")
display(plot(t,θ̈₄,linewidth=2,xaxis="time (s)",yaxis="α (rad/s^2)",
    label="α₄",color="orange",ylim=[0,0.0025]))
savefig("./plots/ex3-33.png")
```

2. Python code for performing the position analysis of four-bar mechanism

```python
# -*- coding: utf-8 -*-
"""
Created on Sun Oct 17 16:45:12 2021

@author: AYUSH
"""
import math
import matplotlib.pyplot as pt
import numpy as np


# fixed parameters
AB=5
BC=6
CD=5
DA=6
theta1=math.pi

#time limit
tspan=[0,45]
teval=np.linspace(tspan[0], tspan[1], 4501)

#independent coordinate
zetain=math.pi/6 # angle subtended by the A-frame [rad]
beta=90*math.pi/(180*45)
theta2=zetain+beta*teval

x=-AB*np.cos(theta2)-DA*np.cos(theta1)
y=-AB*np.sin(theta2)-DA*np.sin(theta1)

z=(np.power(x,2) + np.power(y,2) + CD**2 - BC**2)/(2 * CD)

w=z/(np.sqrt(np.power(x,2) + np.power(y,2)))
theta4=np.arccos(-w) + np.arctan(y/x)

y_C=CD*np.sin(theta4)
y_B=AB*np.sin(theta2)

x_C=CD*np.cos(theta4)+DA
x_B=AB*np.cos(theta2)

theta3=np.arctan(y_C - y_B,x_C - x_B)

#plotting
pt.plot(teval,theta2,color='red')
pt.show()
pt.plot(teval,theta4,color='blue')
pt.show()
```